october 2023

Commacode for noForth T

Loading the assembler in order to define one or two words of your program in assembler takes relatively a lot of space.
COMMACODE produces 'comma code' for the most recently defined word and makes it easy to compile a low level forth word without the use of an active assembler.

```
: COMMACODE   ( -- )
    created lfa> @+ dup >r                \ body doer r: doer
    over < cr if ." routine "
              else ." code " then
    created lfa>n count 1F and type 3 spaces    \ name
    begin @+ u. ." , "
          dup r@ = if  cr ." code> "  then      \ adr = doer ?
          dm 40 hor < if  cr  then  space
    dup here < 0= until
    ." end-code"  cr  r> 2drop ;
```

How to use it

1. include the assembler in noForth
2. include the COMMACODE definition (select and copy the code and paste it into your noForth terminal)
3. define your word in assembler
4. execute the word COMMACODE

Example:

```
code 2@ ( a -- lo hi )
    day  tos 4 #) ldr,       \ @lo
    tos  tos ) ldr,          \ @hi
    day  sp -) str,          \ lo
next, end-code
```

The output will be:

```
code 2@   681B685D ,  600D3904 ,  CA10C804 ,  46A7 ,  end-code
```

✦ Paste the output into your program.

Forth addresses in COMMACODE

It is not a good idea to have absolute addresses of variables, values, subroutines, etc. in commacode. Replace them with their forth names. noForth t uses this method to handle these, start with:
`DATA> ..data field.. CODE>`
The W register contains the address where the data starts.

Example : An interrupt routine that uses a value

```
0 value SUM
routine TALLY ( -- )    \ 1 sum +!
  DATA>  adr sum ,  CODE>
  day  w ) ldr,     \ adr sum
  moon  day ) ldr,  \ sum
  moon 1 # adds,    \ 1 +
  moon  day ) str,  \ to sum
  lr bx,  end-code  commacode
```

```
routine TALLY   E001467A ,  21008EAC ,  682F6815 ,
    602F3701 ,  590B4770 ,  end-code
```

✦ Replace the address of the value with "ADR SUM".

```
adr sum . ↵ 21008EAC  OK.0
```

```
routine PLUS   E001467A ,  adr sum ,  682F6815 ,
 602F3701 ,  10044770 ,  end-code
```